unittest
{
	assert( [ "5 1", "1 0 0 1 0" ].parse.expand.solve == "1 2 2 1 2" );
	assert( [ "5 2", "1 0 0 1 0" ].parse.expand.solve == "3 3 4 4 3" );
	
	assert( [ "200000 200000", "0".repeat( 200000 ).join( " " ) ].parse.expand.solve == "200000".repeat( 200000 ).join( " " ) );		// 追加ケース
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nk = input.front.split.to!( long[] );
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( nk[ 1 ], as );
}

auto solve( long k, long[] as )
{
	// 電球の光量は毎操作ごとに必ず強くなる かつ 電球の数と同じになったら変化がなくなる
	// a^2ぐらいの勢いで強くなるのでO(N^2)実装でも間に合う
	foreach( i; 0L .. k )
	{
		// いもす法で操作後の電球の光量を計算
		auto ts = new long[ as.length + 1 ];
		foreach( j, a; as )
		{
			ts[ max( 0L,               j.signed - a     ) ]++;
			ts[ min( as.length.signed, j.signed + a + 1 ) ]--;
		}
		auto m = 0L;
		foreach( j, ref a; as )
		{
			ts[ j + 1 ] += ts[ j ];
			a = ts[ j ];
			m += ( a == as.length );
		}
		
		// 全電球の光量が最大になったら打ち切り
		if( m == as.length ) break;
	}
	return as.to!( string[] ).join( " " );
}
